
SurrealEx
    

Surreal DB Elixir Library.
Installation
def deps do
  [
    {:surreal_ex, "~> 0.2.0"}
  ]
end
Config
Define a connector file for your project:
defmodule SurrealExTest.Conn do
  use SurrealEx.Conn
end
Adds on your config file, the session configuration parameters:
...
config :surreal_ex, SurrealExTest.Conn,
  interface: :http,
  uri: "http://localhost:8000",
  ns: "testns",
  db: "testdb",
  user: "root",
  pass: "root"
...
SQL Queries
Using your connector you can execute a queries for example:
  test "Creating twice the same example record and we expect catch the error", _state do
    query = "CREATE team:valenciacf SET fullname = 'Valencia Club de Fútbol, S.A.D.', shortname = 'Valencia', founded = 1919, league = 'La Liga';"

    {:ok, response} = Conn.sql(query)
    assert response.status == "OK"
    assert response.detail == nil
    assert !is_list(response.result)

    assert response.result["id"] == "team:valenciacf"
    assert response.result["fullname"] == "Valencia Club de Fútbol, S.A.D."
    assert response.result["shortname"] == "Valencia"
    assert response.result["founded"] == 1919
    assert response.result["league"] == "La Liga"

    {:ok, response} = Conn.sql(query)
    assert response.status == "ERR"
    assert response.detail == "Database record `team:valenciacf` already exists"

  end

Quick CRUD
  defmodule TableBook do
    use SurrealEx.HTTP.Table,
      conn: SurrealExTest.CRUDHttp.TableTest.Conn,
      table: "book"

      # It will allow you use the following methods:
      #
      # get(id) 
      # create(obj)
      # put(id, obj)
      # update(id, obj)
      # delete(id)

  end
(Learn more on our docs.)
Custom Queries
[image: custom_query_execution_flow]
(Learn more on our docs.)
Register & Login
	You requires prepare your SCOPE on SurrealDB for example: 

---define SCHEMAFULL and PERMISSIONS
DEFINE TABLE user SCHEMAFULL
  PERMISSIONS
    FOR select, update WHERE id = $auth.id,
    FOR create, delete NONE;
--- define FIELD's
DEFINE FIELD user ON user TYPE string;
DEFINE FIELD pass ON user TYPE string;
DEFINE FIELD email ON user TYPE string;
DEFINE FIELD role ON user TYPE int;
--- define INDEX's
DEFINE INDEX idx_user ON user COLUMNS user UNIQUE;

-- define SCOPE
DEFINE SCOPE allusers
  -- the JWT session will be valid for 14 days
  SESSION 14d
  -- The optional SIGNUP clause will be run when calling the signup method for this scope
  -- It is designed to create or add a new record to the database.
  -- If set, it needs to return a record or a record id
  -- The variables can be passed in to the signin method
  SIGNUP ( CREATE user SET user = $user, email = $email, role = $role, pass = crypto::argon2::generate($pass))
  -- The optional SIGNIN clause will be run when calling the signin method for this scope
  -- It is designed to check if a record exists in the database.
  -- If set, it needs to return a record or a record id
  -- The variables can be passed in to the signin method
  SIGNIN ( SELECT * FROM user WHERE user = $user AND crypto::argon2::compare(pass, $pass) )
  -- this optional clause will be run when calling the signup method for this scope
	After you will be able to register new users using:

{:ok, _token} = Conn.register("admin", "1234", "example@mail.com")
Or 
user_register = %{
  "user" => "admin",
  "pass" => "1234",
  "email" => "example@mail.com"
  "otherfield1" => ...
  "otherfield2" => ...
}

{:ok,_token} = Conn.register(user_register)
	You can login (and take token) with:

{:ok, token} = Conn.login("admin", "1234")
	And restore information user with:

{:ok, user} = Conn.get_user_by_token(token)
#assert user.email == "example@mail.com"
(Learn more on our docs.)



  

    
Quick CRUD
    

We can define quickly a CRUD for a table using SurrealEx.HTTP.Table. 
The configuration requires the following arguments:
	conn: a SurrealEx connector.
	table: CRUD Table 

Configuration example
defmodule SurrealExTest.CRUDHttp.TableTest do
  use ExUnit.Case

  defmodule Conn do
    use SurrealEx.Conn
  end

  defmodule TableBook do
    use SurrealEx.HTTP.Table,
      conn: SurrealExTest.CRUDHttp.TableTest.Conn,
      table: "book"

      # It will allow you use the following methods:
      #
      # get(id) 
      # create(obj)
      # put(id, obj)
      # update(id, obj)
      # delete(id)
      
  end

Usage example
  test "Creating item with autogen_id & custom id", _state do
    favorite_book = %{
      isbn: "0-7167-1045-5",
      title: "Computers and Intractability: A Guide to the Theory of NP-Completeness",
      author: "Michael R. Garey and David S. Johnson",
      publication_date:	1979
    }

    assert TableBook.get(favorite_book.isbn) == {:not_found, nil}

    {:create, autogen_id} = TableBook.create(favorite_book)

    assert autogen_id != favorite_book.isbn
    "book:" <> autogen_id = autogen_id

    {:found, book} = TableBook.get(autogen_id)

    assert book.isbn == favorite_book.isbn
    assert book.title == favorite_book.title
    assert book.author == favorite_book.author
    assert book.publication_date == favorite_book.publication_date

    idkey = "0-7167-1045-5"
    {:create_or_update, book} = TableBook.put(idkey, favorite_book)

    assert book.id == "book:⟨0-7167-1045-5⟩"
    assert book.isbn == favorite_book.isbn
    assert book.title == favorite_book.title
    assert book.author == favorite_book.author
    assert book.publication_date == favorite_book.publication_date

  end
  test "Create & Read & Update & Delete", _state do
    favorite_book = %{
      isbn: "9780511804090",
      title: "Computational Complexity: A Modern Approach",
      author: "Sanjeev Arora and Boaz Barak",
      publication_date:	2009
    }

    {:not_found, _} = TableBook.get(favorite_book.isbn)
    {:create_or_update, book} = TableBook.put(favorite_book.isbn, favorite_book)

    assert book[:keywords] == nil

    book_keywords_field_json = %{keywords: ["complexity", "NP-hard", "NP-complete"]}
    {:create_or_update, book} = TableBook.update(favorite_book.isbn, book_keywords_field_json)
    assert book[:keywords] != nil

    {:found, book} = TableBook.get(favorite_book.isbn)

    assert book.id == "book:9780511804090"
    assert book.isbn == favorite_book.isbn
    assert book.title == favorite_book.title
    assert book.author == favorite_book.author
    assert book.publication_date == favorite_book.publication_date
    assert book.keywords == ["complexity", "NP-hard", "NP-complete"]

    {:delete, nil} = TableBook.delete(favorite_book.isbn)

  end



  

    
Custom Queries
    

[image: alt text]
	Filtering (Optional): @callback filters(arg :: any) :: [(any() -> {:ok, nil} | {:error, bitstring()})]
During filtering we apply over input a list of functions (any() -> {:ok, nil} | {:error, bitstring()}) that will take the input arguments and will return :ok, or :error with error message. 
Only in case all the the filters will fails (i.e. all will answers with {:ok, ...}) then will start before stage.

	Before (Optional): @callback before(arg :: any) :: {:ok, any()} | {:error, any()} 
On before, we can prepare the arguments that will needed the query, and also we can avoid the execution returning {:error, ...}.
Only in case that before stage returns {:ok, args}, the flows will execute the query.

	Query: @callback query(arg :: any) :: String.t()
On query method we can focus on write our Surreal DB SQL.

	After (Optional):   
	@callback ok(response :: any) :: any()
In case the execution will be succesfull for all the responses (internally we used SurrealEx.Response.all_status_ok?/1) the flow will execute the ok(response).

	@callback error(response :: any) :: any()
In case the execution will be unsuccess for all the responses (internally we used SurrealEx.Response.all_status_ok?/1) the flow will execute error(response).




Example
  defmodule ExampleFlow do
    use SurrealEx.Query,
      conn: SurrealExTest.QueryFiltersTest.Conn

    # its optional
    def filters(_args) do
      [
        ArgsChecker.required(:price_max),
        ArgsChecker.should_be(:price_max, :float),
        ArgsChecker.greater_than(:price_max, 0.0)
      ]
    end

    # before.. as you see on test its optional

    # query callback is mandatory.
    def query(args) do
      "SELECT * FROM car WHERE price <= #{args.price_max}"
    end

    # its optional
    def ok(response) do
      {:ok, response.result}
    end

    # its optional
    def error(_response) do
      {:error, "not expected case"}
    end

  end
Filtering process return us, string errors.
  test "We expected that on before function checks args." do
    args = %{}
    {:error, detail} = ExampleFlow.run(args)

    assert detail == [
      "'price_max' is required",
      "'price_max' should be float",
      "'price_max' should be greater than 0.0"
    ]

    args = %{
      price_max: "15000.00"
    }
    {:error, detail} = ExampleFlow.run(args)

    assert detail == [
      "'price_max' should be float",
      "'price_max' should be greater than 0.0"
    ]
  end
We can see success execution test.
  test "Simple query testing" do
    args = %{
      price_max: 15000.00
    }
    {:ok, list_cars} = ExampleFlow.run(args)

    assert Enum.map(list_cars, fn car -> car["price"] <= 15000 end)
      |> Enum.all?
  end



  

    
Register and Login
    

	You requires prepare your SCOPE on SurrealDB for example: 

---define SCHEMAFULL and PERMISSIONS
DEFINE TABLE user SCHEMAFULL
  PERMISSIONS
    FOR select, update WHERE id = $auth.id,
    FOR create, delete NONE;
--- define FIELD's
DEFINE FIELD user ON user TYPE string;
DEFINE FIELD pass ON user TYPE string;
DEFINE FIELD email ON user TYPE string;
DEFINE FIELD role ON user TYPE int;
--- define INDEX's
DEFINE INDEX idx_user ON user COLUMNS user UNIQUE;

-- define SCOPE
DEFINE SCOPE allusers
  -- the JWT session will be valid for 14 days
  SESSION 14d
  -- The optional SIGNUP clause will be run when calling the signup method for this scope
  -- It is designed to create or add a new record to the database.
  -- If set, it needs to return a record or a record id
  -- The variables can be passed in to the signin method
  SIGNUP ( CREATE user SET user = $user, email = $email, role = $role, pass = crypto::argon2::generate($pass))
  -- The optional SIGNIN clause will be run when calling the signin method for this scope
  -- It is designed to check if a record exists in the database.
  -- If set, it needs to return a record or a record id
  -- The variables can be passed in to the signin method
  SIGNIN ( SELECT * FROM user WHERE user = $user AND crypto::argon2::compare(pass, $pass) )
  -- this optional clause will be run when calling the signup method for this scope
Using Connector
	Define a connector file for your project:

defmodule SurrealExTest.Conn do
  use SurrealEx.Conn
end
Adds on your config file, the session configuration parameters:
...
config :surreal_ex, SurrealExTest.Conn,
  interface: :http,
  uri: "http://localhost:8000",
  ns: "testns",
  db: "testdb",
  user: "root",
  pass: "root"
...
	After you will be able to register new users using:

{:ok, _token} = Conn.register("admin", "1234", "example@mail.com")
Or 
user_register = %{
  "user" => "admin",
  "pass" => "1234",
  "email" => "example@mail.com"
  "otherfield1" => ...
  "otherfield2" => ...
}

{:ok,_token} = Conn.register(user_register)
	You can login (and take token) with:

{:ok, token} = Conn.login("admin", "1234")
	And restore information user with:

{:ok, user} = Conn.get_user_by_token(token)
#assert user.email == "example@mail.com"
Using SurrealEx.HTTPAuth
	After you will be able to register new users using:

{:ok, _token} = SurrealEx.HTTPAuth.register(config, "admin", "1234", "example@mail.com")
Or 
user_register = %{
  "user" => "admin",
  "pass" => "1234",
  "email" => "example@mail.com"
  "otherfield1" => ...
  "otherfield2" => ...
}

{:ok,_token} = SurrealEx.HTTPAuth.register(config, user_register)
	You can login (and take token) with:

{:ok, token} = SurrealEx.HTTPAuth.login(config, "admin", "1234")
	And restore information user with:

{:ok, user} = HTTPAuth.get_user_by_token(config, token)
#assert user.email == "example@mail.com"

Prepare connection configuration for HTTP with Basic Authorization

  
  examples

  
  Examples


iex> config = SurrealEx.Config.for_http("http://localhost:8000","testns","testdb","root","root")
iex> config.kind
:for_http
iex> config.uri
"http://localhost:8000"
iex> config.ns
"testns"
iex> config.db
"testdb"


iex> config = SurrealEx.Config.for_http("http://localhost:8000","testns","testdb","root","root")
iex> config._prepare
%{
  headers: [
    "NS": "testns",
    "DB": "testdb",
    "Authorization": "Basic cm9vdDpyb290",
    "Accept": "application/json"
  ]
}

Documentation for SurrealEx.Conn.
Design Notes: It is interesting can pipe actions (even diferente SQLs) and
can control easily the flow depending if the database answer was sucessfully or not.
  config
  |> sql(..query..)
  |> when_ok_sql(..query..)
  |> when_ok(..callback..)
  |> when_error(..callback..)

all_status_ok? reads all responses status searching that all will be "OK".

  
  example

iex> raw_body = "[{\"time\":\"77.182µs\",\"status\":\"OK\",\"detail\":\"\"}, {\"time\":\"77.182µs\",\"status\":\"OK\",\"detail\":\"\"}]"
iex> {:ok, response} = SurrealEx.Response.build_all({:ok,raw_body})
iex> SurrealEx.Response.all_status_ok?(response)
true

iex> raw_body = "[{\"time\":\"77.182µs\",\"status\":\"ERR\",\"detail\":\"Database record `team:valenciacf` already exists\"}, {\"time\":\"77.182µs\",\"status\":\"OK\",\"detail\":\"\"}]"
iex> {:ok, response} = SurrealEx.Response.build_all({:ok,raw_body})
iex> SurrealEx.Response.all_status_ok?(response)
false

From raw Surreal DB response (that can gives us many responses) we can build the responses list [%SurrealEx.Response{...}] using build_all(raw_body).

  
  example

  
  Example:


iex> raw_body = "[{\"time\":\"77.182µs\",\"status\":\"ERR\",\"detail\":\"Database record `team:valenciacf` already exists\"}]"
iex> {:ok, response} = SurrealEx.Response.build_all({:ok,raw_body})
iex> response
[
  %SurrealEx.Response{
    time: "77.182µs",
    status: "ERR",
    detail: "Database record `team:valenciacf` already exists",
    result: []
  }
]

SurrealEx.Response structure constructor cast the SurrealDB´S JSON responses.

  Example:


iex> json_res = %{ "time" => "775.74µs",
...>  "status" => "OK",
...>  "detail" => "",
...>  "result" => [%{
...>    "id" => "person:bob",
...>    "name" => "bob"
...>  }]
...>}
iex> response = SurrealEx.Response.new(json_res)
iex> response.time
"775.74µs"
iex> response.status
"OK"
iex> response.detail
""
iex> response.result
%{"id" => "person:bob","name" => "bob"}

  Using to_dot_syntax we will can work using dot syntax with result field.
  IMPORTANT: Remember that if you try access to field with dot and this field isnt exist will throws KeyError (See following example...)

  
  example

  
  Example:


iex> json_res = %{ "time" => "127.74µs",
...>  "status" => "OK",
...>  "detail" => "",
...>  "result" => [
...>    %{ "id" => "city:esvalencia", "name" => "Valencia",  "population" => 791_413 },
...>    %{ "id" => "city:esmadrid", "name" => "Madrid", "area" => 604.45 },
...>    %{ "id" => "city:esbarcelona", "name" => "Barcelona" }
...>  ]}
iex> response = SurrealEx.Response.new(json_res)
...>  |> SurrealEx.Response.to_dot_syntax()
iex> [valencia, madrid, barcelona] = response.result
iex> valencia.population
791_413
iex> madrid.area
604.45
iex> barcelona.area
** (KeyError) key :area not found in: %{id: "city:esbarcelona", name: "Barcelona"}

